EE236: Experiment 8

Sheel F. Shah, 19D070052

April 9, 2022

Overview of the experiment

0.1 Aim of the experiment

The aim of this experiment was to understand the workings and characteristics of BJTs.

0.2 Report Pattern

Instead of following the template, I have split the report into sections based on the questions/simulations. Each section is based on one question/simulation, and all associated details are in that section only.

1 Circuit Diagram



Figure 1: Circuit diagram used for all the parts. The values have changes, which can be seen from the code.

2 BJT Parameters ...

2.1 ... in CE

Netlist:

```
Sheel Shah 19D070052

.include bc547.txt

q0 0 1 2 bc547a
r2 3 2 100
i_source 0 1
v_ce 3 0

.dc v_ce 0 100 0.1 i_source 0 1m 0.1m
.control
run

plot -i(v_ce) vs v(2)

.endc
.end
```



Figure 2: I_c vs V_{ce}

Results:

$$\begin{aligned} \beta &= I_c/I_b = 17.9mA/0.1mA = 179.9 \\ reverse\beta &= I_c/I_b = 0.73mA/01mA = 7.3 \\ \alpha &= \beta/(\beta+1) = 0.995 \\ V_A &= 74V \end{aligned}$$

2.2 ... in CB

Netlist:

```
Sheel Shah 19D070052

.include bc547.txt

q0 2 0 1 bc547a
r1 1 3 100
v2 3 0
i_e 2 0

.dc v2  -3.5 100 0.1 i_e 0 10m 1m

.control
run
plot  -i(v2) vs v(1)

.endc
.end
```



Figure 3: I_c vs V_{ce}

Results:

$$\begin{aligned} \beta &= 177 \\ reverse\beta &= 7.1 \\ \alpha &= 0.994 \end{aligned}$$

3 Gummel plot

Netlist:

```
Sheel Shah 19D070052

.include bc547.txt


q0 3 2 1 bc547a
v_be 2 0 0
v_cb 3 2 4
v_dummy 1 0 0

.dc v_be 0.3 3 0.01

.control
run

plot log(i(v_dummy) + i(v_cb)) log(-i(v_cb)) vs v(2)
plot (-i(v_cb))/(i(v_dummy) + i(v_cb)) vs log(-i(v_cb))

.endc
.end
```

Initially in the gummel plot, the difference between I_c and I_b increases, and hence β_{DC} increases. Later as base-emitter voltage increases, I_c and I_b come closer together, and hence β_{DC} decreases.



Figure 4: Collector and base currents against base emitter voltage at a fixed collector to base bias voltage.



Figure 5: β_{dc} vs I_c

4 Small signbal parameters

Netlist:

```
Sheel Shah 19D070052

.include bc547.txt

q0 3 2 0 bc547a
r1 1 2 100k
r2 3 4 1k
v_cc 4 0 9.5
v_bb 1 0

.dc v_bb 0 5 0.1

.control
run


meas dc v_bb1 find v(1) when i(v_cc) = -4.5m
meas dc v_bb2 find v(1) when v(3) = 5
meas dc i_b find i(v_bb) when i(v_cc) = -4.5m
let i_b = -i_b

** beta:
print 4.5m/i_b
** gm:
print 4.5m/25.8m
** r_pi:
print (4.5m/i_b)/(4.5m/25.8m)
** ro:
print 74/4.5m

.endc
.end
```

Results:

$$\beta = 1.974030e + 02$$
$$g_m = 1.744186e - 01$$
$$r_\pi = 1.131777e + 03$$
$$r_o = 1.644444e + 04$$

5 Switching Behaviour

Netlist:

```
Sheel Shah 19D070052

.include bc547.txt
.include 2N3904.txt
.include BAT54.txt

q1 1 2 0 bc547a
*x1 2 1 bat54
*q1 1 2 0 2n3904c
r_b 2 3 1k
r_c 1 4 1k
v_cc 4 0 5
v_in 3 0 pulse(0 5 0 0 0 0.5u 10u)
*vin 3 0 pulse(0 5 0 0 0 0.5m 1m)
*vin 3 0 pulse(0 5 0 0 0 5u 10u)
*.tran 0.01u 3m 1m
.tran 0.001u 1.1m 1m
.control
run
plot v(3) v(1)
meas tran fall trig v(1) val=4.550153e+00 fall=2 targ v(1) val=0.68 fall=2
meas tran storage trig v(3) val=0.0001 rise=2 targ v(1) val=4.550153e+00 fall=2
.endc
.end
```



Figure 6: Results

6 Experiment completion status

I was able to complete all parts of the experiment.