EE236: Experiment 6

Sheel F. Shah, 19D070052

March 16, 2022

Overview of the experiment

0.1 Aim of the experiment

The aim of this experiment was to understand the workings and characteristics of NMOSFETs along with non-idealities. We also understood the CMOS inverter with its transient response.

0.2 Report Pattern

Instead of following the template, I have split the report into sections based on the questions/simulations. Each section is based on one question/simulation, and all associated details are in that section only.



Figure 1: Circuit used for NMOS. The same circuit has been used for parts 1, 2, 3, and the values of the voltages have changed, as can be seen in the code



Figure 2: Circuit used for CMOS. The same circuit has been used for parts 2, 3, 4, and the parameters of the PMOS and NMOS have been changed, as mentioned in the code

1 $I_d - V_{ds}$ characteristics

Netlist used:

```
19D070052 Sheel Shah I_d vs V_ds

.include ALD1105N.txt

** 1 2 3 4: drain gate source body
m1 1 2 0 4 ALD1105N
v_dd 4 0 0
v_id 10 1 0
v_ds 10 0
v_gs 2 0


.dc v_ds 0 5 0.1 v_gs 2.5 4 0.5

* start control
.control
set color0 = rgb:f/f/e
set color1 = rgb:1/1/1

run

plot i(v_id) vs v(10)

** rds by seeing dx/dy near origin:
** 2.5: 1248.18, 3: 1013.56, 3.5: 827.737, 4: 682.482

** r_0 by seeing dx/dy in saturation:
** 2.5: 47674, 3: 25735, 3.5: 19919, 4: 12904

** early voltage:
** sat dy/dx = 7.74935e-05, x0 = 4.05455, y0 = 0.00273448
** c = y-mx = 0.00273448 - 7.74935e-05*(4.05455) = 2.420279e-03
** v_a = -c/m = -3.12320e+01
** 2.5: -31, 3: -29, 3.5: -34, 4: -32
```

```
.endc

.end
```



Figure 3: IV Characteristics

Calculation and values of R_{ds}, R_0, V_A have been mentioned in code

2 V_t and gm measurement

2.1 Linear Region

```
19D070052 Sheel Shah I_d vs V_ds

.include ALD1105N.txt

** 1 2 3 4: drain gate source body
m1 1 2 0 4 ALD1105N
v_dd 4 0 0
v_id 10 1 0
v_ds 10 0 0.2
v_gs 2 0


.dc v_gs 0 5 0.1

* start control
.control
set color0 = rgb:f/f/e
set color1 = rgb:1/1/1

run

plot i(v_id) vs v(2)

** v_t by extrapolating linear region: 0.78
** gm = 9.16118e-05
.endc

.end
```

All measured values are mentioned in the code



Figure 4: I vs V plot in linear region

2.2 Saturation Region

```
19D070052 Sheel Shah I_d vs V_ds

.include ALD1105N.txt

** 1 2 3 4: drain gate source body
m1 1 2 0 4 ALD1105N
v_dd 4 0 0
v_id 10 1 0
v_ds 10 0 5
v_gs 2 0


.dc v_gs 0 5 0.1

* start control
.control
set color0 = rgb:f/f/e
set color1 = rgb:1/1/1

run

plot (i(v_id)) vs v(2)

** v_t by linear region's intercept: 0.706
** K = 2 * slope * slope = 5.107080e-04 A/V^2
.endc

.end
```

All measured values are mentioned in the code



Figure 5: I vs V plot, $\sqrt{|I|}$ vs V plot in saturation region

3 Effect of body bias

```
19D070052 Sheel Shah I_d vs V_ds

.include ALD1105N.txt

** 1 2 3 4: drain gate source body
m1 1 2 0 4 ALD1105N
v_dd 4 0
v_id 10 1 0
v_ds 10 0 0.2
v_gs 2 0


.dc v_gs 0 5 0.1 v_dd -4 0 1

* start control
.control
set color0 = rgb:f/f/e
set color1 = rgb:1/1/1

run

plot i(v_id) vs v(2)

** v_t by extrapolating linear region:
** 0: 0.784615,1: 1.16923, 2: 1.46154, 3: -1.67077, 4: -1.87692
** v_t increases in magnitude as v_sb increases in magnitude

** 0.784615 = v_to
** 1.87692 = 0.784615 + gamma(sqrt(4.9) - sqrt(0.9))
** gamma = 0.86
.endc

.end
```



Figure 6: I vs V_{gs} plot as V_{sb} changes
Measurements and calculations are mentioned in the code

4 CMOS Inverter Characteristics

```
19D070052 Sheel Shah I_d vs V_ds

.include CMOS.txt

m_p 3 1 2 2 cmosp  L=0.4u W=60u AS=4.8e-11 PS=121.6u AD=4.8e-11 PD=121.6u
m_n 3 1 0 0 cmosn L=0.4u W=30u AS=2.4e-11 PS=61.6u AD=2.4e-11 PD=61.6u
v_dd 2 0 3.3
v_in 1 0
c0 3 0 0.05p


.dc v_in 0 3.3 0.01

* start control
.measure dc v_t find v(1) when v(3)=v(1)
.print v_t
** v_t = 1.608554e+00

.control
set color0 = rgb:f/f/e
set color1 = rgb:1/1

run
plot v(3) vs v(1)

.endc
.end
```



Figure 7: V_{out} vs V_{in} plot
Measurement of $V_{threshold}$ is mentioned in the code

5 CMOS Inverter Characteristics ...

5.1 ... as W_p/W_n changes

```
19D070052 Sheel Shah I_d vs V_ds

.include CMOS.txt

m_p1 31 1 2 2 cmosp  L=0.4u W=60u AS=4.8e-11 PS=121.6u AD=4.8e-11 PD=121.6u
m_n1 31 1 0 0 cmosn L=0.4u W=30u AS=2.4e-11 PS=61.6u AD=2.4e-11 PD=61.6u
c1 31 0 0.05p

m_p2 32 1 2 2 cmosp L=0.4u W=60u AS=4.8e-11 PS=121.6u AD=4.8e-11 PD=121.6u
m_n2 32 1 0 0 cmosn L=0.4u W=60u AS=4.8e-11 PS=121.6u AD=4.8e-11 PD=121.6u
c2 32 0 0.05p

m_p3 33 1 2 2 cmosp  L=0.4u W=30u AS=2.4e-11 PS=61.6u AD=2.4e-11 PD=61.6u
m_n3 33 1 0 0 cmosn L=0.4u W=60u AS=4.8e-11 PS=121.6u AD=4.8e-11 PD=121.6u
c3 33 0 0.05p

v_dd 2 0 3.3
v_in 1 0

.dc v_in 0 3.3 0.01

* start control
.measure dc v_t1 find v(1) when v(31)=v(1)
.print v_t1
.measure dc v_t2 find v(1) when v(32)=v(1)
.print v_t2
.measure dc v_t3 find v(1) when v(33)=v(1)
.print v_t3

** v_t1 = 1.608554e+00
** v_t2 = 1.392646e+00
** v_t3 = 1.203754e+00

.control
```

```
set color0 = rgb:f/f/e
set color1 = rgb:1/1

run
plot v(31) vs v(1) v(32) vs v(1) v(33) vs v(1)
.endc
.end
```



Figure 8: V_{out} vs V_{in} plot

The curve becomes sharper as the width ratio increases. The switching threshold also drops as the width ratio increases.

5.2 ... as V_{dd} changes

```
19D070052 Sheel Shah I_d vs V_ds

.include CMOS.txt

m_p1 31 1 21 21 cmosp  L=0.4u W=60u AS=4.8e-11 PS=121.6u AD=4.8e-11 PD=121.6u
m_n1 31 1 0 0 cmosn L=0.4u W=30u AS=2.4e-11 PS=61.6u AD=2.4e-11 PD=61.6u
v_dd1 21 0 1.5
v_in 1 0
c1 31 0 0.05p

m_p2 32 1 22 22 cmosp  L=0.4u W=60u AS=4.8e-11 PS=121.6u AD=4.8e-11 PD=121.6u
m_n2 32 1 0 0 cmosn L=0.4u W=30u AS=2.4e-11 PS=61.6u AD=2.4e-11 PD=61.6u
v_dd2 22 0 3
v_in 1 0
c2 32 0 0.05p


.dc v_in 0 3.3 0.01

* start control

.control
set color0 = rgb:f/f/e
set color1 = rgb:1/1

run
plot v(31) vs v(1) v(32) vs v(1)

.endc
.end
```



Figure 9: V_{out} vs V_{in} plot
The curve is very sharp for 1.5V supply voltage, as compared to 3V supply voltage.

6 CMOS Inverter propagation delay ...

6.1 ... as W_p/W_n changes

```
19D070052 Sheel Shah I_d vs V_ds

.include CMOS.txt


m_p0 30 2 1 1 cmosp L=0.4u W=60u AS=4.8e-11 PS=121.6u AD=4.8e-11 PD=121.6u
m_n0 30 2 0 0 cmosn L=0.4u W=30u AS=2.4e-11 PS=61.6u AD=2.4e-11 PD=61.6u
c0 30 0 0.05p

m_p1 31 2 1 1 cmosp L=0.4u W=60u AS=4.8e-11 PS=121.6u AD=4.8e-11 PD=121.6u
m_n1 31 2 0 0 cmosn L=0.4u W=60u AS=4.8e-11 PS=121.6u AD=4.8e-11 PD=121.6u
c1 31 0 0.05p

m_p2 32 2 1 1 cmosp L=0.4u W=30u AS=2.4e-11 PS=61.6u AD=2.4e-11 PD=61.6u
m_n2 32 2 0 0 cmosn L=0.4u W=60u AS=4.8e-11 PS=121.6u AD=4.8e-11 PD=121.6u
c2 32 0 0.05p

v_dd 1 0 3.3V
v_in 2 0 PULSE (0 3.3 0 0.02n 0.02n 4n 8n)


.tran 0.7p 8n


** t_r0                 =  4.665267e-11 targ=  4.096014e-09 trig=  4.049361e-09
** t_f0                 =  3.674940e-11 targ=  6.793967e-11 trig=  3.119027e-11
** t_r1                 =  5.675018e-11 targ=  4.110246e-09 trig=  4.053496e-09
** t_f1                 =  2.372796e-11 targ=  4.880620e-11 trig=  2.507824e-11
** t_r2                 =  8.996286e-11 targ=  4.153922e-09 trig=  4.063959e-09
** t_f2                 =  1.992199e-11 targ=  4.229081e-11 trig=  2.236883e-11
** t_phl0               =  3.523537e-11 targ=  4.523537e-11 trig=  1.000000e-11
** t_plh0               =  4.407193e-11 targ=  4.064072e-09 trig=  4.020000e-09
** t_phl1               =  2.368525e-11 targ=  3.368525e-11 trig=  1.000000e-11
** t_plh1               =  5.213147e-11 targ=  4.072131e-09 trig=  4.020000e-09
```

```
** t_phl2             =  1.933043e-11 targ=  2.933043e-11 trig=  1.000000e-11
** t_plh2             =  7.511498e-11 targ=  4.095115e-09 trig=  4.020000e-09
** (t_phl0 + t_plh0)/2 = 3.465475e-11
** (t_phl1 + t_plh1)/2 = 3.290935e-11
** (t_phl2 + t_plh2)/2 = 4.222314e-11

.control
run

meas tran t_r0 TRIG v(30) VAL=0.33 RISE=1 TARG v(30) VAL=2.97 RISE=1
meas tran t_f0 TRIG v(30) VAL=2.97 FALL=1 TARG v(30) VAL=0.33 FALL=1

meas tran t_r1 TRIG v(31) VAL=0.33 RISE=1 TARG v(31) VAL=2.97 RISE=1
meas tran t_f1 TRIG v(31) VAL=2.97 FALL=1 TARG v(31) VAL=0.33 FALL=1

meas tran t_r2 TRIG v(32) VAL=0.33 RISE=1 TARG v(32) VAL=2.97 RISE=1
meas tran t_f2 TRIG v(32) VAL=2.97 FALL=1 TARG v(32) VAL=0.33 FALL=1

meas tran t_phl0 TRIG v(2) VAL=1.65 RISE=1 TARG v(30) VAL=1.65 FALL=1
meas tran t_plh0 TRIG v(2) VAL=1.65 FALL=1 TARG v(30) VAL=1.65 RISE=1

meas tran t_phl1 TRIG v(2) VAL=1.65 RISE=1 TARG v(31) VAL=1.65 FALL=1
meas tran t_plh1 TRIG v(2) VAL=1.65 FALL=1 TARG v(31) VAL=1.65 RISE=1

meas tran t_phl2 TRIG v(2) VAL=1.65 RISE=1 TARG v(32) VAL=1.65 FALL=1
meas tran t_plh2 TRIG v(2) VAL=1.65 FALL=1 TARG v(32) VAL=1.65 RISE=1

print (t_phl0 + t_plh0)/2
print (t_phl1 + t_plh1)/2
print (t_phl2 + t_plh2)/2

.endc
.end
```

All the required values asked have been mentioned in the code.
We see that rise time increases as width ratio increases, whereas fall time decreases.
The propagation delay first decreases and then increases.

6.2 ... as V_{dd} changes

```
19D070052 Sheel Shah I_d vs V_ds
.include CMOS.txt


v_dd 1 0 3.2
v_in 2 0 PULSE (0 3.2 0 0.02n 0.02n 4n 8n)

m_p 3 2 1 1 cmosp L=0.4u W=60u AS=4.8e-11 PS=121.6u AD=4.8e-11 PD=121.6u
m_n 3 2 0 0 cmosn L=0.4u W=30u AS=2.4e-11 PS=61.6u AD=2.4e-11 PD=61.6u
c0 3 0 0.05p

.tran 0.7p 8n
.control
run
meas tran t_phl TRIG v(2) VAL=1.65 RISE=1 TARG v(3) VAL=1.65 FALL=1
meas tran t_plh TRIG v(2) VAL=1.65 FALL=1 TARG v(3) VAL=1.65 RISE=1
print (t_phl+t_plh)/2

** 2: 6.254763e-11, 2.2: 5.326970e-11, 2.4: 4.729788e-11, 2.6: 4.308321e-11,
** 2.8: 3.993140e-11, 3: 3.747619e-11, 3.2: 3.550326e-11

.endc
.end
```

All the required values asked have been mentioned in the code.



Figure 10: Propagation delay vs supply voltage.
The plot looks like that of $1/x^2$.

7 Experiment completion status

I was able to complete all parts of the experiment.